▪ Plot Figure 1

In[1]:=
```
SetDirectory["~/writing/WIP/KappaLib/"];
<< kappaLib.m
```

```
KappaLib v1.1
```

▪ Fresnel surface

In[3]:=
```
Ax = - DiagonalMatrix[{1, 2, 3}];
Bx = DiagonalMatrix[{1, 1, 1}];
Cx = 0 IdentityMatrix[3];
Dx = 0 IdentityMatrix[3];

kappa = emABCDToKappa[Ax, Bx, Cx, Dx];
```

In[8]:=
```
xi = {xi0, xi1, xi2, xi3};
fresnel = Simplify[emKappaToFresnel[kappa, xi]];
```

In[10]:=
```
FullSimplify[fresnel]
```

Out[10]=
$$-6\,\mathrm{xi0}^4-\left(\mathrm{xi1}^2+\mathrm{xi2}^2+\mathrm{xi3}^2\right)\left(\mathrm{xi1}^2+2\,\mathrm{xi2}^2+3\,\mathrm{xi3}^2\right)+\mathrm{xi0}^2\left(5\,\mathrm{xi1}^2+8\,\mathrm{xi2}^2+9\,\mathrm{xi3}^2\right)$$

In[11]:=
```
Simplify[Solve[(fresnel /. xi0 → 1) == 0, xi3]]
```

Out[11]=
$$\left\{\left\{\mathrm{xi3}\to-\frac{\sqrt{9-4\,\mathrm{xi1}^2-5\,\mathrm{xi2}^2-\sqrt{4\,\mathrm{xi1}^4+4\,\mathrm{xi1}^2\left(-3+\mathrm{xi2}^2\right)+\left(3+\mathrm{xi2}^2\right)^2}}}{\sqrt{6}}\right\},\right.$$
$$\left\{\mathrm{xi3}\to\frac{\sqrt{9-4\,\mathrm{xi1}^2-5\,\mathrm{xi2}^2-\sqrt{4\,\mathrm{xi1}^4+4\,\mathrm{xi1}^2\left(-3+\mathrm{xi2}^2\right)+\left(3+\mathrm{xi2}^2\right)^2}}}{\sqrt{6}}\right\},$$
$$\left\{\mathrm{xi3}\to-\frac{\sqrt{9-4\,\mathrm{xi1}^2-5\,\mathrm{xi2}^2+\sqrt{4\,\mathrm{xi1}^4+4\,\mathrm{xi1}^2\left(-3+\mathrm{xi2}^2\right)+\left(3+\mathrm{xi2}^2\right)^2}}}{\sqrt{6}}\right\},$$
$$\left.\left\{\mathrm{xi3}\to\frac{\sqrt{9-4\,\mathrm{xi1}^2-5\,\mathrm{xi2}^2+\sqrt{4\,\mathrm{xi1}^4+4\,\mathrm{xi1}^2\left(-3+\mathrm{xi2}^2\right)+\left(3+\mathrm{xi2}^2\right)^2}}}{\sqrt{6}}\right\}\right\}$$

In[12]:=
```
sol[xi1_, xi2_, sigma_] :=
```
$$\frac{1}{\sqrt{6}}\left(\sqrt{\left(9-4\,\mathrm{xi1}^2-5\,\mathrm{xi2}^2+\mathrm{sigma}\sqrt{4\,\mathrm{xi1}^4+4\,\mathrm{xi1}^2\left(-3+\mathrm{xi2}^2\right)+\left(3+\mathrm{xi2}^2\right)^2}\right)}\right)$$

▪ Check that solutions parameterise the Fresnel surface

In[13]:=
```
Simplify[fresnel /. {xi0 → 1, xi3 → sol[xi1, xi2, +1]}]
Simplify[fresnel /. {xi0 → 1, xi3 → sol[xi1, xi2, -1]}]
```

Out[13]= 0

Out[14]= 0

▪ Define PlotRange

In[15]:=
```
pRange = {{-1, Sqrt[3] + 1}, {-1, Sqrt[3] + 1}, {-1, Sqrt[3] + 1}};
pPoints = 80;
width = 0.003;
```

Compute curves in xz-plane

In[18]:=
```
Solve[sol[x, 0, 1] == 0, x]
Solve[sol[x, 0, -1] == 0, x]
```

Out[18]= $\{\{x \to -\sqrt{3}\}, \{x \to \sqrt{3}\}\}$

Out[19]= $\{\{x \to -\sqrt{2}\}, \{x \to \sqrt{2}\}\}$

In[20]:=
```
P1 = ParametricPlot3D[{xi1, 0, sol[xi1, 0, 1]},
    {xi1, 0, Sqrt[3]}, Axes → False, Boxed → False, PlotRange → pRange,
    PlotPoints → pPoints, PlotStyle → {Blue, Thickness[width]}];
P2 = ParametricPlot3D[{xi1, 0, sol[xi1, 0, -1]},
    {xi1, 0, Sqrt[2]}, Axes → False, Boxed → False,
    PlotPoints → pPoints, PlotStyle → {Blue, Thickness[width]}];
```

Compute curves in yz-plane

In[22]:=
```
Solve[sol[0, y, 1] == 0, y]
Solve[sol[0, y, -1] == 0, y]
```

Out[22]= $\{\{y \to -\sqrt{3}\}, \{y \to \sqrt{3}\}\}$

Out[23]= $\{\{y \to -1\}, \{y \to 1\}\}$

In[24]:=
```
P3 = ParametricPlot3D[{0, y, sol[0, y, 1]},
    {y, 0, Sqrt[30]}, Axes → False, Boxed → True,
    PlotPoints → pPoints, PlotStyle → {Blue, Thickness[width]}];
P4 = ParametricPlot3D[{0, y, sol[0, y, -1]},
    {y, 0, 10}, Axes → False, Boxed → False,
    PlotPoints → pPoints, PlotStyle → {Blue, Thickness[width]}];
```

Find curves in xy-plane

In[26]:=
```
fresnel /. {xi3 → 0, xi0 → 1}
```

Out[26]= $-6 + 5\,\mathrm{xi1}^2 - \mathrm{xi1}^4 + 8\,\mathrm{xi2}^2 - 3\,\mathrm{xi1}^2\,\mathrm{xi2}^2 - 2\,\mathrm{xi2}^4$

In[27]:=
```
Solve[% == 0, xi2]
```

Out[27]= $\left\{\left\{\mathrm{xi2} \to -\frac{\sqrt{2-\mathrm{xi1}^2}}{\sqrt{2}}\right\}, \left\{\mathrm{xi2} \to \frac{\sqrt{2-\mathrm{xi1}^2}}{\sqrt{2}}\right\}, \left\{\mathrm{xi2} \to -\sqrt{3-\mathrm{xi1}^2}\right\}, \left\{\mathrm{xi2} \to \sqrt{3-\mathrm{xi1}^2}\right\}\right\}$

In[28]:=
```
P5 = ParametricPlot3D[{xi1, Sqrt[2 - xi1^2]/Sqrt[2], 0},
    {xi1, 0, Sqrt[2]}, Axes → False, Boxed → True,
    PlotPoints → pPoints, PlotStyle → {Blue, Thickness[width]}];

P6 = ParametricPlot3D[{xi1, Sqrt[3 - xi1^2], 0},
    {xi1, 0, Sqrt[3]}, Axes → False, Boxed → False,
    PlotPoints → pPoints, PlotStyle → {Blue, Thickness[width]}];
```

Solve singular point

In[30]:=
```
(* Find singular points on Fresnel surface*)
FrNabla = {
     D[fresnel, xi[[1]]] == 0,
     D[fresnel, xi[[2]]] == 0,
     D[fresnel, xi[[3]]] == 0,
     D[fresnel, xi[[4]]] == 0
    } /. xi0 → 1;
```

In[31]:=
```
FullSimplify[FrNabla] // MatrixForm
```

Out[31]//MatrixForm=

$$\begin{pmatrix} 5\,xi1^2 + 8\,xi2^2 + 9\,xi3^2 == 12 \\ xi1\left(-5 + 2\,xi1^2 + 3\,xi2^2 + 4\,xi3^2\right) == 0 \\ xi2\left(-8 + 3\,xi1^2 + 4\,xi2^2 + 5\,xi3^2\right) == 0 \\ xi3\left(-9 + 4\,xi1^2 + 5\,xi2^2 + 6\,xi3^2\right) == 0 \end{pmatrix}$$

In[32]:=
```
(* there are more singular points in the complex domain,
but let us only solve real singular points *)
Solve[FrNabla, {xi1, xi2, xi3}, Reals]
```

Out[32]=

$$\left\{\left\{xi1 \to -\sqrt{\frac{3}{2}},\ xi2 \to 0,\ xi3 \to -\frac{1}{\sqrt{2}}\right\},\ \left\{xi1 \to -\sqrt{\frac{3}{2}},\ xi2 \to 0,\ xi3 \to \frac{1}{\sqrt{2}}\right\},\right.$$
$$\left.\left\{xi1 \to \sqrt{\frac{3}{2}},\ xi2 \to 0,\ xi3 \to -\frac{1}{\sqrt{2}}\right\},\ \left\{xi1 \to \sqrt{\frac{3}{2}},\ xi2 \to 0,\ xi3 \to \frac{1}{\sqrt{2}}\right\}\right\}$$

In[33]:=
```
(* The fresnel equation is mirror symmetric across all
  xi_i planes. Let us therefore assume that xi_i >0 for all i. *)
```

$$sPoint = \left\{\sqrt{\frac{3}{2}},\ 0,\ \frac{1}{\sqrt{2}}\right\};$$

■ Check that the solutions intersect at the singular point

In[34]:=
```
sol[Sqrt[3 / 2], 0, 1]
sol[Sqrt[3 / 2], 0, -1]
```

Out[34]= $\frac{1}{\sqrt{2}}$

Out[35]= $\frac{1}{\sqrt{2}}$

In[36]:=
```
singPoint = Graphics3D[ {AbsolutePointSize[5.5], Point[{sPoint}]}];
```

■ Coordinate axes

In[37]:=
```
eps = 0.5;
back = +0.35;
w = 0.04;
```

In[40]:=
```
xArr = Graphics3D[{
     Line[{{0, 0, 0}, {Sqrt[3] + eps, 0, 0}}],
     Line[{{Sqrt[3] + back, +w, 0}, {Sqrt[3] + eps, 0, 0}}],
     Line[{{Sqrt[3] + back, -w, 0}, {Sqrt[3] + eps, 0, 0}}]
    }];

yArr = Graphics3D[{
     Line[{{0, 0, 0}, {0, Sqrt[3] + eps, 0}}],
     Line[{{w, Sqrt[3] + back, 0}, {0, Sqrt[3] + eps, 0}}],
     Line[{{-w, Sqrt[3] + back, 0}, {0, Sqrt[3] + eps, 0}}]
    }];

zArr = Graphics3D[{
     Line[{{0, 0, 0}, {0, 0, Sqrt[2] + eps}}],
     Line[{{0, w, Sqrt[2] + back}, {0, 0, Sqrt[2] + eps}}],
     Line[{{0, -w, Sqrt[2] + back}, {0, 0, Sqrt[2] + eps}}]
    }];
```

- **Show all curves on surface**

In[43]:=
```
plot = Show[{P1, P2, P3, P4, P5, P6, singPoint, xArr, yArr, zArr, singPoint}]
```

Out[43]=



- **The image in the paper was further manipulated with image manipulation softwares.**